
Frequency of Midazolam Dosage by Nursing Shift

Web Appendix: Software Code from SAS and WinBUGS for Models in Article 
 
1. Frequentist ZIP Model in SAS (assumes vertical dataset where each row is a person-
 observation) 
 
*zero inflated poisson, ; 
titl  'NLMIXe ED of ZIP Model for Midazolam';  
proc nlmixed data=temC.Midaz3Vert; 
       parameters  b1=0 b2=0 b3=0 b4=0 b5=0 b6=0 b7=0 b8=0 b9=0 b10=0 b11=0  
               a1=0 a2=0 a3=0 a4=0 a5=0 a6=0 a7=0 a8=0 a9=0 a10=0 a11=0 s2u=1 
; 
 
   /* linear predictor for the inflation probability      */ 
   linpinfl = a1 + a2*shift + a3*night + a4*nonwhite + a5*male + a6*iqcodeA +
 a7*intubated + a8*APS  
        + a9*icudeath + a10*age + a11*LOSlt5d ; 
    ; 
   /* infprob = inflation probability for zeros           */ 
   /*         = logistic transform of the linear predictor*/ 
   infprob  = 1/(1+exp(-linpinfl)); 
   /* Poisson mean */ 
   lambda   = exp(b1 + b1R + b2*shift + b3*night + b4*nonwhite + b5*male + 
b6*iqcodeA + b7*intubated + b8*APS  
        + b9*icudeath5 + b10*age + b11*LOSlt5d ); 
   /* Build the ZIP log likelihood */ 
   if MidazDose=0 then  
        ll = log(infprob + (1-infprob)*exp(-lambda)); 
   else ll = log((1-infprob)) - lambda  + MidazDose*log(lambda) - 
lgamma(MidazDose + 1); 
   model MidazDose ~ general(ll); 
   random b1R  ~ normal(0,s2u) subject=studyid; 
 
run; 
 
 
 
 
2. Bayesian ZIP Model in WinBUGS (assumes horizontal dataset with single row per person) 
 
 
model 
{ 
             
################################################################# 
## we nest the 15 shifts within each studyid among the cohort of ICU survivors 
##who were in ICU >= 3 days and received Midazolam during that time 
## the Midazolam doses have been rounded to nearest unit for use with Poisson  
## this is the classic mixture form of the ZIP model where occurrences of zeroes 
## are modeled with a logistic regression and the positive integers with a Poisson 
## and final likelihood is a mix of the two distributions 
 
 
   for(studyid in 1 : N) { 
    for(shift in 1 : NumShifts[studyid]) {  # a column called NumShifts is person specific 
    
   # logistic likelihood for extra zeroes 
   # coefficients centered around the sample mean of study population to facilitate convergence 

 logExtraZ[studyid,shift] <-     beta[1]         
    + beta[2]*(shift - 8) + 
    + beta[3]*(night[shift] - 0.33) 
    + beta[4]*(nonwhite[studyid]  - 0.17) 
    + beta[5]*(male[studyid]  - 0.55) 
    + beta[6]*(iqcodea[studyid]  - 0.3) 
    + beta[7]*(intubated[studyid]  - 0.75) 
    + beta[8]*(aps[studyid]   - 24.3) 
    + beta[9]*(icudeath[studyid] - 0.17) 
    + beta[10]*(age[studyid]  - 73.4) 
    + beta[11]*(losLT5d[studyid]  - 0.29) 
    + Rint[studyid, 1]   # this is the random intercept for each person in logistic model 
      
     
      
 probZ[studyid,shift] <- 1/(1 + exp(-logExtraZ[studyid,shift]))  # probZ = probability of a zero dose 
      
             
   # Poisson likelihood for positive integers 
 
 log(lambda[studyid,shift]) <-   alpha[1]  
    + alpha[2]*(shift - 8) 
    + alpha[3]*(night[shift] - 0.33) 
    + alpha[4]*(nonwhite[studyid]  - 0.17) 
    + alpha[5]*(male[studyid]  - 0.55) 
    + alpha[6]*(iqcodea[studyid]  - 0.3) 
    + alpha[7]*(intubated[studyid]  - 0.75) 
    + alpha[8]*(aps[studyid]   - 24.3) 
    + alpha[9]*(icudeath[studyid] - 0.17) 
    + alpha[10]*(age[studyid]  - 73.4) 
    + alpha[11]*(losLT5d[studyid]  - 0.29) 
    + Rint[studyid, 2]  # this is the random intercept for each person in Poisson model 
              
              
 probGTZ[studyid,shift] <- probZ[studyid,shift]*(1 - exp(-lambda[studyid,shift])) # prob of MidazDose  > 0   
      
 d[studyid,shift] <- step(MidazDose[studyid,shift] - 1) 
        
      
# Log Likelihood is a Mixture of logistic and Poisson 
      
LogLLH[studyid,shift] <-  (1 -  d[studyid,shift])*log(1 - probGTZ[studyid,shift])    # for all zero doses 
   + d[studyid,shift]*(log(probGTZ[studyid,shift]) + MidazDose[studyid,shift]*log(lambda[studyid,shift]) 
   - lambda[studyid,shift] - loggam(MidazDose[studyid,shift] + 1) - log(1 - exp(- lambda[studyid,shift])))   
       
   # following lines employ the zeroes trick, i.e., WinBUGS’ equivalent of general() in SAS  
   K <- 10000  # constant required for "zeros trick"   
   zeros[studyid,shift] <- 0 
   zeros[studyid,shift] ~ dpois(phi[studyid,shift]) 
   phi[studyid,shift] <- -LogLLH[studyid,shift] + K   
                                                                           
                                                                                                
  }  # close shift loop 
  }  # close studyid loop 
################################################################# 
 
#################################################  
   
      
  # prior distributions for fixed  model terms 
    
   for (i in 1:11) 
    { 
    alpha[i] ~ dnorm(0,0.1) 
    } 
     
    for (i in 1:11) 
    { 
    beta[i] ~ dnorm(0,0.1) 
    } 

####################################################################    
    # prior distributions for random terms 
    
   # first for normal locations 
   for (i in 1:113) 
    { 
    Rint[i,1] ~ dnorm(0,tau1) 
    m[i] <- psi*Rint[i,1] 
    Rint[i,2] ~ dnorm(m[i], tau2) 
    }   
     
    psi~dnorm(0, 0.001) 
    tau1 ~ dgamma(0.1,0.1) 
    tau2 ~ dgamma(0.1,0.1) 
     
   # second for variances 
     
    sigma1 <- 1/sqrt(tau1)   # SD of Rint[,1] 
    denom1 <- pow(psi,2)/tau1 + 1/tau2 
    sigma2 <- sqrt(denom1)  # SD of Rint[,2] 
    denom2 <- sqrt(denom1/tau1) 
    num <- psi/tau1 
    rho <- num/denom2  # corr(Rint[,1],Rint[,2]) 
     
######################################################################      
    } 
####################################################################  
    
    

  
3. Bayesian Random Effects Poisson Model in WinBUGS (assumes horizontal dataset with 
single row per person) 
 
 
 
 
model 
{ 
    
           
################################################################# 
## we nest the 15 shifts within each studyid among the cohort of ICU survivors 
##who were in ICU >= 3 days and received Midazolam during that time 
 
 
   for(studyid in 1 : N) { 
   for(shift in 1 : NumShifts[studyid]) { # a column called NumShifts is person specific 
    
   # Poisson likelihood for observed doses 
   MidazDose[studyid,shift] ~ dpois(mu[studyid,shift])  
   log(mu[studyid,shift]) <-   alpha[1]  
      + alpha[2]*(shift - 8) 
      + alpha[3]*(iqcodea[studyid]  - 0.3) 
      + alpha[4]*(nonwhite[studyid]  - 0.17) 
      + alpha[5]*(male[studyid]  - 0.55) 
      + alpha[6]*(night[shift] - 0.33) 
      + alpha[7]*(intubated[studyid]  - 0.75) 
      + alpha[8]*(aps[studyid]   - 24.3) 
      + alpha[9]*(icudeath[studyid] - 0.17) 
      + alpha[10]*(age[studyid]  - 73.4) 
      + alpha[11]*(losLT5d[studyid]  - 0.29) 
      + theta[studyid] + perShift[studyid,shift] # random effects person-shift 

          } 
          } 
################################################################# 
 
 # Prior distribution for the person-shift intercepts 
  for(studyid in 1 : N) { 
   for(shift in 1 : 15) { 
     perShift[studyid,shift] ~dnorm(0,tau.perShift) 
     } 
     } 
 
 
 # Prior distribution for the random intercepts 
   for (studyid in 1:N) 
    {  
    theta[studyid]~dnorm(0,tau.theta) 
    } 
        
#################################################  
 
          
  # prior distributions for the model coefficients 
    
   alpha[1:11]~dmnorm(alphamu[],Sigma[,]) 
     
     
####################################################################    
  #Hyperprior distributions on inverse variance parameter of random effects 
    
     tau.theta  ~ dgamma(1,1) 
    tau.perShift ~  dgamma(1,1) 
     
    } 
######################################################################      
     
####################################################################  
    
    

  


